
*=============The_Clipping_AM_Spectrum===========================

One would expect that clipping an AM signal should result in a 
pure square wave. It is possible to print out parts of the 
spectrum so that one no longer has to assume.

AM_Clipping
* dsauersanjose@aol.com  1/18/20
* www.idea2ic.com                 replace(OPT-SPACE)=>SPACE
*    ___
*   |OUT|   .5sidebands should product +/- 1.0radians PM
*   |___|   .3sidebands should product +/- 0.6radians PM
*    _|_    +/- (.6/6.28318)*100ms =   +/- 9.5msec
*   /_  \
*  // \  \
*  \   \//
*   \___/
*    _|_
*    ///
*
*=========The_SideBand_Tests=========================
VT        VT     0      PWL(   0      0      2       2 )   DC     0
VP        VP     0      DC     3.141592653589793
V2P       V2P    0      DC     6.283185307179586
BREF      VREF   0      V =    cos( v(V2P)*10*v(VT))
BAM1      VAM1   0      V =    cos( v(V2P)*10*v(VT)) +.4*cos( v(V2P)*8*v(VT)) +.4*cos( v(V2P)*12*v(VT)) 
BAM2      VAM2   0      V =    cos( v(V2P)*10*v(VT)) -.4*sin( v(V2P)*8*v(VT)) +.4*sin( v(V2P)*12*v(VT))
*================================================================

The two different types of Amplitude modulation are shown.

*=========Clipping_Circuit=========================
BREFC     VREFC  0      V =    2*u( v(VREF))-1
BAM1C     VAM1C  0      V =    2*u( v(VAM1))-1
BAM2C     VAM2C  0      V =    2*u( v(VAM2))-1
*=========Run_Sim=========================
.control 
set       pensize = 2 
*TRAN     TSTEP  TSTOP  TSTART TMAX   ?UIC?
tran      1m     2      0       1m
*=========PlotResults============================================
plot      vam1    vam2 
plot      vam1c   vam2c
*================================================================

After clipping, all wave forms look like square waves.

*===========Find_Spectrum_VAM1=========================
linearize 



set         specwindow =    "rectangular" 
*SPEC       FSTART  FSTOP    FSTEP   VECTOR
spec        .5      500     .5       v(VAM1) 
plot        mag(v(VAM1))     loglog  ylimit 1m 1
*================================================================

With the sidebands set to 40% of the carrier, the
amplitude modulation will be at 80%.

 

Before clipping, the +/- side bands exist.

*===========ReRun_Simulation==================================
*TRAN       TSTEP  TSTOP  TSTART TMAX   ?UIC?
tran        1m     2      0      1m
*============================================================

Transient needs to be rerun to get VAM1C.

*===========Find_Spectrum_VAM1==================================
linearize 
set         specwindow =    "rectangular" 
*SPEC       FSTART  FSTOP    FSTEP   VECTOR
spec        .5      500     .5       v(VAM1C) 
plot        mag(v(VAM1C))   loglog  ylimit 1m 1
*================================================================

It appears there is just a pure square wave.
Better print out the spectrum to be sure.

*=========Print_Spectrum_VAM1C=========================
foreach   ii 10 20 30 40 50 60 70
let i =   $ii*2 -1 
let fr=   frequency[i]
let       vfftr = real(v(VAM1C)[i])
let       vffti = imag(v(VAM1C)[i])



echo      index= $&i   freq= $&fr   real=  $&vfftr  imag=  $&vffti
end
*================================================================

The following print out of the spectrum confirms the 
expectations. 

index = 19  freq = 10 real =  1.27282    imag = -0.0179152
index = 39  freq = 20 real =  0.00057728 imag = -7.88908E-17
index = 59  freq = 30 real = -0.423156   imag = 0.0179152
index = 79  freq = 40 real = -0.00057728 imag = 3.37699E-16
index = 99  freq = 50 real =  0.25255    imag = -0.0179152
index = 119 freq = 60 real =  0.00057728 imag = 2.1878E-16
index = 139 freq = 70 real = -0.17895    imag = 0.0179152

===========Full_Netlist_For_Copy_Paste=======================
AM_Clipping
* dsauersanjose@aol.com  1/18/20
* www.idea2ic.com                 replace(OPT-SPACE)=>SPACE
*    ___
*   |OUT|   .5sidebands should product +/- 1.0radians PM
*   |___|   .3sidebands should product +/- 0.6radians PM
*    _|_    +/- (.6/6.28318)*100ms =   +/- 9.5msec
*   /_  \
*  // \  \
*  \   \//
*   \___/
*    _|_
*    ///
*
*V_PULSE# NODE_P NODE_N PULSE( VINIT  VPULSE TDELAY TRISE  TFALL  PWIDTH PERIOD )
VIN       VIN    0      PULSE( 0      1      .25    1u     1u     .5     1      )
*=========The_SideBand_Tests=========================
VT        VT     0      PWL(   0      0      2       2 )   DC     0
VP        VP     0      DC     3.141592653589793
V2P       V2P    0      DC     6.283185307179586
BREF      VREF   0      V =    cos( v(V2P)*10*v(VT))
BAM1      VAM1   0      V =    cos( v(V2P)*10*v(VT)) +.4*cos( v(V2P)*8*v(VT)) +.4*cos( v(V2P)*12*v(VT)) 
BAM2      VAM2   0      V =    cos( v(V2P)*10*v(VT)) -.4*sin( v(V2P)*8*v(VT)) +.4*sin( v(V2P)*12*v(VT))
*=========Clipping_Circuit=========================
BREFC     VREFC  0      V =    2*u( v(VREF))-1
BAM1C     VAM1C  0      V =    2*u( v(VAM1))-1
BAM2C     VAM2C  0      V =    2*u( v(VAM2))-1
*=========Run_Sim=========================
.control 
set       pensize = 2 
*TRAN     TSTEP  TSTOP  TSTART TMAX   ?UIC?
tran      1m     2      0       1m
*=========PlotResults============================================
plot      vam1    vam2 
plot      vam1c   vam2c
*=========Find_Spectrum_VAM1=========================
linearize 
set       specwindow =    "none" 
*SPEC     FSTART  FSTOP    FSTEP   VECTOR
spec      .5      500     .5       v(VAM1) 
plot      mag(v(VAM1))   loglog   ylimit 1m 1
*=========ReRun_Simulation=========================
tran      1m     2      0       1m
*=========Find_Spectrum_VAM1C=========================
linearize 
set       specwindow =    "none" 
*SPEC     FSTART  FSTOP    FSTEP   VECTOR
spec      .5      500     .5       v(VAM1C) 
plot      mag(v(VAM1C))   loglog  ylimit 1m 1
*=========Print_Spectrum_VAM1C=========================
foreach   ii 10 20 30 40 50 60 70
let i =   $ii*2 -1 
let fr=   frequency[i]
let       vfftr = real(v(VAM1C)[i])
let       vffti = imag(v(VAM1C)[i])
echo      index= $&i   freq= $&fr   real=  $&vfftr  imag=  $&vffti
end
.endc 
.end

